Understanding Functions 

You can find three types of functions within the JAWS® scripting language. The three types of functions include: 

· Built-in

· User-defined

· Event 

Built-in functions are included with the JAWS scripting language and cannot be modified. User-defined functions are functions created for a specific task within a script file and can be modified. Event functions are part of JAWS and can be modified. A description of each type of function follows.

You can find complete descriptions of all built-in, user-defined and event functions that come with JAWS by reviewing the Freedom Scientific Developer Network function reference manual. A copy of the manual is provided on the Script Writing with JAWS course CD. You can also download the manual from the following URL:

http://www.freedomscientific.com/fs_support/doc_scriptfunction.asp
Built-In Functions

Freedom Scientific has provided over 1,000 pre-defined functions within the JAWS scripting language. Each function performs a specific series of activities. You use these functions as the building blocks or instructions used to develop your scripts. For example, the RouteJAWSToPC function checks the position of the JAWS and PC cursors, makes the necessary calculations and moves the JAWS cursor to the same point as the PC cursor. All the activity is transparent to the JAWS user.

There are built-in functions that accomplish such tasks as:  speaking messages, retrieving specific blocks of text, navigating the windows hierarchy and more. You can use the Insert Function dialog to insert any built-in function into the body of your scripts. You can also type the name of the function manually into the body of your script. If you choose to type the built-in function into your script manually, you must make sure you spell the name of the function correctly. You must also be sure the parameters the function requires are of the correct type and in the correct order.

User-Defined Functions

You can also use the script language to create or modify user-defined functions. The difference between a built-in function and a user-defined function is that you can modify user-defined functions while you cannot modify built-in functions. 

A user-defined function is a self-contained module of scripting code that you develop to accomplish a specific task. A function is useful when you need to accomplish the same task from a variety of scripts and other user-defined functions. You can use the code of the function without having to write it multiple times. When  you write the code once in a function, the time you need to maintain the code is reduced. Rather than modifying blocks of code in several different locations, you need only to make the changes in one place. After you have made the changes, re-compile any scripts or user-defined functions  that call the newly modified function.

After you have created the function, you can call it from another script or function, and the task is performed. When the function completes processing, control is returned to the calling script or function where processing continues. You can find many examples of user-defined functions written by the script developers at Freedom Scientific in the default script file, Default.jss. To view the default script file in the Script Manager, do the following:

1. Press insert+f2 to activate the Run JAWS Manager dialog.

2. Press S to select the Script Manager followed by enter to start the manager.

3. Press ctrl+shift+d to open the default script file.

Event Functions

The purpose of event functions is to cause JAWS to perform specific tasks automatically without waiting for you to initiate the action with a keystroke. Each event function is triggered by its corresponding windows event. For example, when an application writes new text to the screen, the new text event is triggered. This event causes JAWS to perform the NewTextEvent function to determine what type of text was written to the screen. Event functions differ from user-defined functions in that you cannot create new event functions. However, you can modify the code in existing event functions.

All parameters for an event function are declared between the left and right parentheses in the function begin statement. When JAWS calls this event function, values are automatically passed for each of the parameters between the parentheses of the function call statement. You cannot add or remove parameters that an event function uses. However since values for each of these parameters are passed when the function is called, you can use the data provided by any of them when you need to customize the event function. 

A list of some of the most important event functions follows:

· AutoStartEvent

· FocusChangedEvent

· NewTextEvent

JAWS performs the AutoStartEvent function when an application starts and any time the application becomes active. Any code that you do not place within a conditional statement is performed each time the function is called. You can use the AutoStartEvent function to initialize any global variables used by the application script file, speak an introductory message when the application starts or perform functions to initialize other settings used by the scripts and user-defined functions contained in the application-specific script file.
Note:
Using a global variable to determine if the AutoStartEvent has occurred once is a useful technique. This technique is used when you want the AutoStartEvent function to perform a specific task the first time an application starts but not every time the application receives focus.

JAWS performs the FocusChangedEvent function when focus moves between applications, dialogs, and dialog controls. Each time a focus change occurs, the FocusChangedEvent compares the previous window with the current window and determines exactly what JAWS should speak.

For example, pressing tab to move from control to control within a dialog box triggers the FocusChangedEvent. In turn, JAWS performs the FocusChangedEvent function to determine the type of change and what JAWS should speak. So when you move from an edit box to a combo box, JAWS speaks the associated prompt of the combo box, the text of “Combo box” and any value contained in the combo box.

JAWS performs the NewTextEvent function when any new text is written to the screen. The function determines what type of text is written and what is spoken by JAWS. For example, pressing Shift+down arrow to select lines of text in your favorite text editor causes the NewTextEvent to occur. JAWS in turn performs the NewTextEvent function to speak the text as it is selected.

There are over 25 event functions JAWS calls automatically. You can review the code of each of these event functions by reviewing the default script file, Default.jss within the Script Manager. 

1. Press insert+f2 to activate the Run JAWS Manager dialog.

2. Press S to select the script Manager followed by enter to start the manager.

3. Press ctrl+shift+d to open the default script file.

4. Press ctrl+l to display the list of all scripts and functions contained within the file. Look for any function name that includes the word “event” in its name. When you find a function containing “event,” press enter to move directly to that function.

Parameters and Returns

The simplest function does not receive or return any data to the calling script or function. A good example of a function that does not require any parameters or return any data is the SayLine built-in function. This function only speaks information and does not require any parameters or return any data to the calling script or function. 

Passing Parameters

A parameter is data you pass to a function. You pass a parameter at run time rather than include it as part of the code because the information may be different each time you call the function. A parameter can be any of the variable types used in the scripting language: handle, integer (int), object or string. You declare a parameter like a variable. Instead of declaring the parameter in the body of your function using the key word Var, you place the type and name of the parameter between the parentheses following the function name. When your function requires multiple parameters, you separate each parameter declaration with a comma. Some examples of function beginning lines containing parameter declarations follow:

Void Function SelectingText(int nMode)

    void Function ProcessSelectText(int nAttributes,string buffer)

You can also use the New Script dialog within the Script Manager to declare parameters. After you have entered all the general information about the new function, press ctrl+tab to move to the Parameters page of the New Script dialog. Press tab to move to the New Parameter edit box and type the name of the parameter. You should follow the same rules for naming parameters as you would for naming variables. Press tab to move to the Description edit box and type in a description of the parameter. After you have entered a description for the new parameter, press tab until you reach the Available Types list box. Use your arrow keys to select the type of data the parameter will contain. After you have chosen a parameter type, move to and activate the add button to add the parameter to your function.

You declare a function that requires a single string parameter as follows:

Void Function SayName (string sMyName)

SayFormattedMessage (OT_MESSAGE, sMyName)
End Function

In the above example, the SayName function receives a string parameter, sMyName, from the calling script or function. The SayFormattedMessage function uses the value in sMyName as its first parameter. You can call the SayName function in one of the following ways:

Example 1:

Script SayMyName ()

SayName ("John")

End Script

Example 2:

Script SayMyName ()

Var

string sMyName

Let sMyName = “John”

SayName (sMyName)

End Script

The data being passed in the sMyName parameter can either be enclosed in quotation marks as in the first example or declared as a variable in the calling script as in the second example. In either case, the function is expecting a string of data to be passed. When you insert the SayName function into the above script using the Insert Function dialog, you are prompted to provide the data for the parameter just as if you were inserting one of the JAWS built-in functions.

Returning Data

A return is data that is sent back from a function to the calling script or function after the function completes processing. You use the keyword, return, to return the data to the calling script or function. The calling script or function uses this returned data for further processing. 

A function can return any of the following:

· Void

· Handle
· Integer

· Object
· String
When you create a function, the return type is placed in front of the key word, Function, in the beginning line of the user-defined function. You select the return type from the New Script dialog within the Script Manager at the same time you create the user-defined function.

A function that requires no parameters and does not return any data to the calling script or function is declared as follows:

Void Function SpeakTheName ()

Var

string sMyName

InputBox ("Enter the name:", "EnterName", sMyName)
; prompt the user for their name

If sMyName != cscNull  Then
; determine if a value was entered in the input box


SayFormattedMessage (OT_MESSAGE, sMyName)
Else


SayFormattedMessage (OT_ERROR, “No name entered.”)

EndIf

End Function

In the above example, the Void keyword precedes the Function keyword. The void keyword indicates that the function does not return any data to the calling script or function.

A function that requires no parameters and returns a string value is declared as follows:

String Function ReturnTheName ()

Var

string sMyName

InputBox ("Enter the name:", "EnterName", sMyName)
; prompt the user for their name

Return sMyName ; return the name to the calling script or function

End Function

In the above example, the String keyword precedes the Function key word. The String keyword indicates that the function returns a string value to the calling script or function.

